Coil Gun Example

This example considers the computation of the force on an iron ball at various positions relative to a wound, air-cored coil.The configuration is similar to some coil guns that are designed to shoot ball bearings.The specific configuration in pictured below. For sanity-check purposes, the force on the ball can also be approximated using a point dipole model. The worksheet plots both the dipole force prediction and the FEA results on the same graph.



There is a "convenient" closed - form expression for the on - axis flux denisty of the coil, where J is the coil current density; ri and ro are the inner and outer radii respectively, l is the axial length, and rs is the radius of the sphere :

$$Bz = \frac{1}{4} J\, uo \left((l - 2 Z)\, \text{Log}\left[\frac{2\, ro + \sqrt{4\, ro^2 + (l - 2 Z)^2}}{2\, ri + \sqrt{4\, ri^2 + (l - 2 Z)^2}}\right] + (l + 2 Z)\, \text{Log}\left[\frac{2\, ro + \sqrt{4\, ro^2 + (l + 2 Z)^2}}{2\, ri + \sqrt{4\, ri^2 + (l + 2 Z)^2}}\right] \right);$$

Parameters for our particular case are :

```
Inch = 0.0254;
```

```
sub = {ri → 0.25 * Inch, ro → 1.25 * Inch, l → 1.5 * Inch,
   J → 10 * 10^6, Nd → 1 / 3, rs → (5 / 32) * Inch, uo → 4. Pi 10^(-7)};
```

And as a check, a plot of the on - axis field is as follows :

```
Plot[Bz /. sub, {Z, -3 Inch, 3 Inch}]
```



Volume of the steel sphere

```
vs = (4 / 3) Pi rs^3;
```

Dipole moment of the steel sphere. Depends on the volume of the sphere, the demagnetization factor of the sphere, and the applied field.

```
m = vs * Bz / (uo * Nd);
```

The force on the sphere can then be obtained via the formula for force on a dipole, which simplifies in the on - axis case to F = m*D[Bz, z]

```
dBz = D[Bz, Z];
```

```
F = m * dBz;
```

```
AnalyticalForcePlot = Plot[F /. sub /. Z → z Inch, {z, -1.5, 0}]
```



```
<< c:\\femm42\\mathfemm\\mathfemm.m
```

```
MathFEMM loaded at Thu 15 Aug 2013 23:54:17
```

```
OpenFEMM[]
```

```
OpenDocument[NotebookDirectory[] <> "coilgun.fem"]
```

```
MISaveAs[NotebookDirectory[] <> "temp.fem"];
```

```
frc = {};
For[k = 0, k ≤ 15, k++,
  MIAnalyze[];
  MILoadSolution[];
  MOGroupSelectBlock[1];
  fz = MOBlockIntegral[19];
  frc = Append[frc, {(k - 15) / 10., fz}];
  MOClose[];
  MISelectGroup[1];
  MIMoveTranslate[0, 0.1];
 ];
```

```
CloseFEMM[]
```

```
frc
```

```
{{-1.5, 0.0532421}, {-1.4, 0.0767361}, {-1.3, 0.11249}, {-1.2, 0.166223},
 {-1.1, 0.249897}, {-1., 0.376777}, {-0.9, 0.552885}, {-0.8, 0.73976},
 {-0.7, 0.847892}, {-0.6, 0.831133}, {-0.5, 0.712849}, {-0.4, 0.576257},
 {-0.3, 0.415264}, {-0.2, 0.277749}, {-0.1, 0.130461}, {0, 0.00232598}}
```

```
FEAForcePlot = ListPlot[frc]
```



```
Show[AnalyticalForcePlot, FEAForcePlot, Frame → True,
 GridLines → Automatic, FrameLabel → {"Position from Coil Center, Inches",
   "Force on Ball, Newtons", "FEA and Analytical Force Results"}, ImageSize → 500]
```